
Разбор задач муниципального этапа олимпиады по информатике

1. Серое королевство (все классы)
Тема: разбор случаев
Сложность: простая

Нужно найти максимум и минимум 3 чисел, а затем в зависимости от их разности вывести одно из
сообщений.

Пример реализации:
uses math;
var r,g,b:integer;
begin
  read(r,g,b);
  if max(max(r,g),b)-min(min(r,g),b)<=25 then
    writeln('ALLOWED')
  else
    writeln('FORBIDDEN');
end.

2. Алгоритм (7-9 класс)
Тема: реализация программы по схеме алгоритма
Сложность: простая

Пример реализации:
var i,n,s,m,x:integer;
begin
  read(n);
  m:=0;
  s:=0;
  for i:=1 to n do
  begin
    read(x);
    s:=s+x;
    if s<0 then s:=0;
    if m<s then m:=s;
  end;
  writeln(m);
end.

2. Раздел королевства (10-11 класс)
Тема: полный перебор, поиск минимума
Сложность: простая

Пример реализации:
uses math;
var x,y:array[1..100] of integer;
  n,i,j,k,mini,minj:integer;
  d,di,dj,mind:extended;
begin
  read(n);
  for i:=1 to n do
    read(x[i],y[i]);
  mind:=1e9;
  for i:=1 to n-1 do { перебирает кандидатов на столицы }
    for j:=i+1 to n do
    begin
      d:=0;
      for k:=1 to n do { для всех городов }
      begin 
        { считаем расстояние до столиц }
        di:=sqrt(sqr(x[i]-x[k])+sqr(y[i]-y[k]));
        dj:=sqrt(sqr(x[j]-x[k])+sqr(y[j]-y[k]));
        { и берем меньшее из них }



        d:=d+min(di,dj);
      end;
      { если результат лучше, чем раньше, то запоминаем результат }
      if mind>d then
      begin
        mind:=d;
        mini:=i;
        minj:=j;
      end;
    end;
  writeln(mini,' ',minj);
end.

3. Робот-маляр (7-8 класс)
Тема: автоматы
Сложность: простая

Пример реализации:
0 R _ > 2
0 G _ > 4
0 B _ > 6
2 _ _ > 3
4 _ _ > 5
6 _ _ > 7
3 ? R < 0
5 ? G < 0
7 ? B < 0

3. Дележ (9 класс)
Тема: математика, двоичное представление чисел
Сложность: простая

Нужно найти последнюю 1 в двоичном представлении числа a.
Пример реализации:

var n,a,b,i:integer;
begin
  read(n,a,b);
  for i:=0 to n do
  begin
    if (a and (1 shl i))<>0 then { проверить i-й бит числа a }
    begin
      writeln(n-i);
      break;
    end;
  end;
end.

3. Дележ-2 (10-11 класс)
Тема: двоичное представление чисел, моделирование
Сложность: ниже среднего

Пример реализации:
var n,a,b,c,i,k,e:integer;
function ibit(x,i:integer):boolean; { вернуть i-й бит числа x }
begin
  ibit:=(x and (1 shl i))<>0;
end;
begin
  read(n,a,b);
  c:=(1 shl n)-a-b; { (1 shl n) = 2^n }
  e:=1; { количество единиц в i-м бите, которые нужно распределить }
  k:=0;
  for i:=n downto 0 do
  begin



    if ibit(a,i) then e:=e-1;
    if ibit(b,i) then e:=e-1;
    if ibit(c,i) then e:=e-1;
    k:=k+e; { неиспользованные биты (куски золота) делим на 2 части } 
    e:=e*2; { для этого потребуется произнести }
  end;
  writeln(k);
end.

4. Пеппи Длинныйчулок (7-9 класс, частичное решение 10-11 класс)
Тема: сортировка по двум критериям, структуры
Сложность: средняя

Пример реализации:
type info=record len1,len2:integer; end;
var len:array[1..1000] of integer;
    n,k,i,j,m:integer;
    p:array[1..500000] of info;
function less(a,b:info):boolean;
begin
  if a.len2-a.len1<b.len2-b.len1 then less:=true
  else if a.len2-a.len1>b.len2-b.len1 then less:=false
  else if a.len1<b.len1 then less:=true
  else less:=false;
end;
procedure ksort(l,r:integer);
var x,t:info;
    i,j:integer;
begin
  if l>=r then exit;
  x:=p[l+random(r-l+1)];
  i:=l;
  j:=r;
  while i<=j do
  begin
    while less(p[i],x) do inc (i);
    while less(x,p[j]) do dec (j);
    if i<=j then
    begin 
      t:=p[i];p[i]:=p[j];p[j]:=t; 
      inc(i); dec(j); 
    end;
  end;
  if k<=j then ksort(l,j);
  if k>=i then ksort(i,r);
end;
begin
  read(n,k);
  for i:=1 to n do
    read(len[i]);
  m:=0;
  for i:=1 to n-1 do
    for j:=i+1 to n do
    begin
      inc(m);
      p[m].len1:=len[i];
      p[m].len2:=len[j];
    end;
  ksort(1,m);
  writeln(p[k].len1,' ',p[k].len2);
end.

4. Пеппи Длинныйчулок-2 (полное решение 10-11 класс)
Тема: бинарный поиск
Сложность: выше среднего



Пример реализации:
type func=function(x:integer):boolean;
var 
  len:array[1..100000] of integer;
  bas,d,dif,n,i,j:integer;
  k:int64;
function bsearch(a,b:integer;f:func):integer;forward;
  
function fun1(x:integer):boolean;
begin
  fun1:=len[x]-len[bas]>dif;
end;
function koldif(x:integer):int64; { количество пар с разностью <=x }
var kol:int64;
  j:integer;
begin
  kol:=0;
  dif:=x;
  for j:=1 to n-1 do 
  begin
    bas:=j;
    kol:=kol+(bsearch(j+1,n+1,@fun1)-j-1);
  end;
  koldif:=kol;
end;
function fun2(x:integer):boolean;
begin
  fun2:=koldif(x)>=k;
end;
function bsearch(a,b:integer;f:func):integer;
var c:integer;
begin
  while a<b do
  begin
    c:=(b+a) div 2;
    if f(c) then b:=c
    else a:=c+1;
  end;
  bsearch:=b;
end;
begin
  read(n,k);
  for i:=1 to n do
    read(len[i]);
  { находим разность, при которой кол-во пар с разностью не больше этой >= k }
  d:=bsearch(1,1000000000,@fun2); 
  { определяем, сколько пар лишних }
  k:=k-koldif(d);
  dif:=d;
  { находим пары с этой разностью, начиная с максимальной длины меньшего чулка }
  for i:=n downto 1 do
  begin
    bas:=i;
    j:=bsearch(i+1,n+1,@fun1);
    if len[j-1]-len[i]=d then
      if k=0 then
      begin
        writeln(len[i],' ',len[j-1]);
        break;
      end
      else
        inc(k);
  end;
end.

